Technical Note No. 10

Note on the Food Consumption Score (FCS) Variable
NOTE ON THE FOOD CONSUMPTION SCORE (FCS) VARIABLE

Background:

The index of diversity and balanced food consumption adopted in this report, called Food Consumption Score (FCS), is the core indicator of consumption recommended by the World Food Programme - Vulnerability Analysis Mapping (WFP-VAM, 2008). It has been used in several report/analysis around the world (see the list in WFP-VAM, pp.6-7, 2008), apart from being the selected indicator for the most recent analysis of the food security in Yemen (WFP, 2012).

This indicator takes into account three dimensions of food consumption: dietary diversity, food frequency, and relative nutritional importance. The score is calculated using the weighted frequency of consumption of nine food groups consumed by a household during the seven days before the survey. The weights for each group of food are listed in the WFP-VAM (2008), and are based on the nutrient density analysis. Higher (lower) weights are assigned to the best (worst) food group’s quality in terms of caloric density, and macro and micro nutrient content (Box 1).
### Box 1. Grouping of food items and their relative nutritional weight

<table>
<thead>
<tr>
<th>Food Groups</th>
<th>Food Items</th>
<th>Weight</th>
</tr>
</thead>
<tbody>
<tr>
<td><strong>1</strong> Main staples</td>
<td>Grains and cereals: rice, maize, sorghum, millet, noodles.</td>
<td>2</td>
</tr>
<tr>
<td></td>
<td>Tubers and roots: potatoes, sweet potatoes, beet roots, etc.</td>
<td></td>
</tr>
<tr>
<td><strong>2</strong> Pulses</td>
<td>Beans, lentils, peas, nuts.</td>
<td>3</td>
</tr>
<tr>
<td><strong>3</strong> Vegetables</td>
<td>Vegetables: spinach, broccoli, other green leaf vegetable.</td>
<td>1</td>
</tr>
<tr>
<td><strong>4</strong> Fruits</td>
<td>Fruits</td>
<td>1</td>
</tr>
<tr>
<td><strong>5</strong> Meat and Fish</td>
<td>Chicken meat, other poultry.</td>
<td>4</td>
</tr>
<tr>
<td></td>
<td>Meat: beef, lamb, goats.</td>
<td></td>
</tr>
<tr>
<td></td>
<td>Eggs</td>
<td></td>
</tr>
<tr>
<td></td>
<td>Seafood: fish, prawns, crabs, etc.</td>
<td></td>
</tr>
<tr>
<td><strong>6</strong> Milk</td>
<td>Dairy products (excluding butter): milk, yogurt, etc.</td>
<td>4</td>
</tr>
<tr>
<td><strong>7</strong> Sugar</td>
<td>Sugar and honey</td>
<td>0.5</td>
</tr>
<tr>
<td><strong>8</strong> Oil</td>
<td>Oils and fats: cooking oils, butter, ghee, etc.</td>
<td>0.5</td>
</tr>
<tr>
<td><strong>9</strong> Condiments</td>
<td>Coffee, tea, juices, beverages.</td>
<td>0</td>
</tr>
</tbody>
</table>


In the NSPMS, the frequency of consumption is based not on a daily frequency, but rather is split into five separate intervals. In order to have the frequency in number/proportion of days, we used the average of each category, as shown in the Box 2:
Box 2. Average of consumption frequencies by week

<table>
<thead>
<tr>
<th>Frequency of consumption</th>
<th>Average of weekly consumption frequency</th>
</tr>
</thead>
<tbody>
<tr>
<td>Daily: 6-7 days per week</td>
<td>6.5</td>
</tr>
<tr>
<td>3-5 times per week</td>
<td>4</td>
</tr>
<tr>
<td>1-2 times per week</td>
<td>1.5</td>
</tr>
<tr>
<td>1-3 times per month</td>
<td>0.07*</td>
</tr>
<tr>
<td>Not at all</td>
<td>0</td>
</tr>
</tbody>
</table>

* In order to have the average of weekly consumption frequency, the average monthly frequency is divided by 30.

Having all these information, Food Consumption Score is calculated according to the following steps:

**Step 1:** sum all the consumption frequencies of food items of the same group, and recode the value of each group above 7 as 7;

**Step 2:** multiply the value obtained for each food group by its weight;

**Step 3:** sum of the weighted food groups scores, thus creating the Food Consumption Score (FCS);

**Step 4:** use the standard thresholds (see Box 3) to recode the continuous variable of food consumption score to a categorical one. Scoring below 28, a household has a diet that does not include at least a staple, oil, sugar, and vegetables on a daily basis.

Box 3. Food consumption profiles

<table>
<thead>
<tr>
<th>FCS</th>
<th>Profiles</th>
</tr>
</thead>
<tbody>
<tr>
<td>&lt; 28</td>
<td>Poor food consumption</td>
</tr>
<tr>
<td>≥ 28 and &lt; 42</td>
<td>Borderline food consumption</td>
</tr>
<tr>
<td>≥ 42</td>
<td>Poor food consumption</td>
</tr>
</tbody>
</table>

Source: WFP (2012), Technical Annex
References

WORLD FOOD PROGRAMME, VULNERABILITY ANALYSIS AND MAPPING (WFP-VAM), (2008). Food consumption analysis: calculation and use of the food consumption score in food security analysis.


ANNEX 1 – Syntax in Stata format

* 44) Index of diversity and balanced food consumption (iii.3699)

local foodvars grain vegetable bean tuber fruit chicken meat egg fish oil dairy sugar coffee

gen grain = p5_q_9_1
gen vegetable = p5_q_9_2
gen bean = p5_q_9_3
gen tuber = p5_q_9_4
gen fruit = p5_q_9_5
gen chicken = p5_q_9_6
gen meat = p5_q_9_7
gen egg = p5_q_9_8
gen fish = p5_q_9_9
gen oil = p5_q_9_10
gen dairy = p5_q_9_11
gen sugar = p5_q_9_12
gen coffee = p5_q_9_13

recode `foodvars' (1 = 6.5) (2 = 4) (3 = 1.5) (4 = 0.07) (5 = 0) (9 = .)
tempvar gr1_staples gr2_pulses gr3_veg gr4_fruit gr5_meat gr6_milk gr7_sugar gr8_oil gr9_condim
egen `gr1_staples' = rowtotal(grain tuber) // possivel erro (corrigido)
gen `gr2_pulses' = bean
gen `gr3_veg' = vegetable
gen `gr4_fruit' = fruit
egen `gr5_meat' = rowtotal(chicken meat egg fish)
gen `gr6_milk' = dairy
gen `gr7_sugar' = sugar
gen `gr8_oil' = oil
gen `gr9_condim' = coffee
replace `gr1_staples' = 7 if `gr1_staples' > 7 & `gr1_staples' != .
replace `gr5_meat' = 7 if `gr5_meat' > 7 & `gr5_meat' != .
drop `foodvars'

tempvar gr1_w gr2_w gr3_w gr4_w gr5_w gr6_w gr7_w gr8_w gr9_w

gen `gr1_w' = `gr1_staples'*2
gen `gr2_w' = `gr2_pulses'*3

gen `gr3_w' = `gr3_veg'*1

gen `gr4_w' = `gr4_fruit'*1

gen `gr5_w' = `gr5_meat'*4

gen `gr6_w' = `gr6_milk'*4

gen `gr7_w' = `gr7_sugar'*0.5

gen `gr8_w' = `gr8_oil'*0.5

gen `gr9_w' = `gr9_condim'*0

egen fcs = rowtotal(`gr1_w' `gr2_w' `gr3_w' `gr4_w' `gr5_w' `gr6_w' `gr7_w' `gr8_w' `gr9_w')

label variable fcs "Food Consumption Score"
gen byte fcs_gr = .
replace fcs_gr = 1 if fcs < 28
replace fcs_gr = 2 if fcs >= 28 & fcs < 42
replace fcs_gr = 3 if fcs >= 42
label variable fcs_gr "Profiles of Food Consumption Score"
label define fcs_gr 1 "severe" 2 "moderate" 3 "food secure"
label values fcs_gr fcs_gr